

MODELICA

Introdução

PME3201 Laboratório de Simulações Numéricas


Prof. Ponge

Exemplos de Sistema Mecânico – Translação

24.11.2017

Exemplo: sistema massa-mola-amortecedor

- Esfera em meio viscoso
- Parâmetros:
 - $m = 10,0 \text{ kg}$
 - $k = 1000,0 \text{ N/m}$
 - $c = 50,0 \text{ kg/s}$
 - $g = 10,0 \text{ m/s}^2$



$$P = m \cdot g$$

$$F_k = k \cdot y$$

$$F_d = c \cdot v$$

$$m \cdot \frac{d v}{d t} = P - F_k - F_d$$

$$\frac{d y}{d t} = v$$


Modelo Plano

```
1 model Sistema_MKC_Plano "Sistema Massa-Mola-Amortecedor - Modelo Plano"
2   constant Real g = 10.0;
3   parameter Real m = 10.0;
4   parameter Real k = 1000.0;
5   parameter Real c = 50.0;
6   Real y(start = 0.0), v(start = 0.0);
7   Real P, Fk, Fd;
8   equation
9     P = m * g;
10    Fk = k * y;
11    Fd - c * v = 0;
12    m * der(v) = P - Fd - Fk;
13    v = der(y);
14 end Sistema_MKC_Plano;
```


Modelo Plano com variáveis físicas

```
1 model Sistema_MKC_Fisico "Sistema Massa-Mola-Amortecedor com variáveis físicas da biblioteca MSL"
2   constant Modelica.SIunits.Acceleration g = 9.80665;
3   parameter Modelica.SIunits.Mass m = 10.0;
4   parameter Modelica.SIunits.TranslationalDampingConstant c = 50.0;
5   parameter Modelica.SIunits.TranslationalSpringConstant k = 1000.0;
6   Modelica.SIunits.Position y(start = 0.0);
7   Modelica.SIunits.Velocity v(start = 0.0);
8   Modelica.SIunits.Force P, Fk, Fd;
9   equation
10    P = m * g;
11    Fk = k * y;
12    Fd - c * v = 0;
13    m * der(v) = P - Fd - Fk;
14    v = der(y);
15 end Sistema_MKC_Fisico;
```

Modelo Gráfico – Biblioteca MSL



Modelo em Diagrama de Blocos



$$\frac{d^2 y}{dt^2} = g - \frac{k}{m} \cdot y - \frac{c}{m} \cdot \frac{d y}{dt}$$
$$\frac{d y}{dt} = v$$

Biblioteca de Componentes

- Class connecter
 - acoplamento
- Class model
 - massa
 - mola
 - amortecedor
 - pontoFixo
 - forcaPeso


Acomplamento

```
1 connector acoplamento "Conector mecânico linear"  
2   Real y;  
3   flow Real f;  
4 end acoplamento;
```

- Variável *entre* - y
 - Compatibilidade
 - tem valor igual para componentes que compartilham o mesmo conector
- Variável *através (flow)* - f
 - Continuidade (equilíbrio)
 - Somam para zero para componentes que compartilha o mesmo conector

Mola

```
1 model mola "Componente mecânico com dois conectores - mola linear"  
2   parameter Real k = 1000.0;  
3   acoplamento a, b;  
4   Real x, F, d;  
5   equation  
6     x = a.y;  
7     F = a.f;  
8     d = a.y - b.y;  
9     F = k * d "Lei de Hooke";  
10    a.f + b.f = 0;  
11 end mola;
```



$$x = a.y$$

$$F = k \cdot d$$


$$F = a.f$$

$$a.f + b.f = 0$$

$$d = a.y - b.y$$

Amortecedor

```
1 model amortecedor "Componente mecânico com dois conectores - amortecedor"
2   parameter Real c = 100.0;
3   acoplamento a, b;
4   Real x, F, d;
5   equation
6     x = a.y;
7     F = a.f;
8     d = a.y - b.y;
9     F = c * der(d) "Amortecimento Viscoso";
10    a.f + b.f = 0;
11 end amortecedor;
```



$$x = a.y$$

$$F = c \cdot v$$


$$F = a.f$$

$$a.f + b.f = 0$$

$$d = a.y - b.y \quad v = \text{der}(d)$$

Massa

```
1 model massa "Componente mecânico com dois conectores - massa"  
2   parameter Real m = 10.0;  
3   acoplamento a, b;  
4   Real x, F, d, v;  
5   equation  
6     x = a.y;  
7     F = a.f;  
8     d = a.y - b.y;  
9     m * der(v) = a.f + b.f "Lei de Newton";  
10    d = 0;  
11    v = der(x);  
12 end massa;
```



$$x = a.y$$

$$F = a.f$$

$$d = a.y - b.y$$


$$m \cdot \text{der}(v) = a.f + b.f$$

$$d = 0$$

$$v = \text{der}(x)$$

Ponto Fixo


```
1 model pontoFixo "Componente mecânico com um conector - ponto fixo"
2   acoplamento a;
3   Real x, F;
4   equation
5     x = a.y;
6     F = a.f;
7     a.y = 0;
8 end pontoFixo;
```



$$x = a.y \quad x = 0$$
$$F = a.f$$

Força Peso

```
1 model forcaPeso "Componente mecânico com um conector - força constante"
2   constant Real g = 10.0;
3   parameter Real m = 100.0;
4   acoplamento b;
5   Real x, P = m * g;
6   equation
7     x = b.y;
8     P + b.f = 0;
9 end forcaPeso;
```




$$x = b.y \quad P + F = 0$$


$$F = b.f$$

Sistema MKC com componentes da biblioteca


```
1 model SistemaMassaMolaAmortecedor "Sistema MKC com biblioteca de componentes"  
2   massa m1(m = 10.0);  
3   mola k1(k = 1000.0);  
4   amortecedor c1(c = 50.0);  
5   pontoFixo p1;  
6   forcaPeso Peso(m = 10.0);  
7 equation  
8   connect(Peso.b, m1.a);  
9   connect(m1.b, k1.a);  
10  connect(m1.b, c1.a);  
11  connect(k1.b, p1.a);  
12  connect(c1.b, p1.a);  
13 end SistemaMassaMolaAmortecedor;
```



Simulação



Simulação



Simulação

